

# Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills  
Synopsys, Inc.  
Maricopa, AZ  
dillan@synopsys.com

Chip Haldane  
The Chip Abides, LLC  
Gilbert, AZ  
chip@thechipabides.com

## Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

## I. CONSTRAINTS AND POLICY CLASS REVIEW

Random objects and constraints are the foundational building blocks of constrained random verification in SystemVerilog. The simplest implementations embed fixed constraints within a class definition. Embedded constraints lack flexibility; all randomized object instances must meet these requirements exactly as they are written.

In-line constraints using the `with` construct offer marginally better flexibility. Although these external constraints allow greater variability of random objects, their definitions are still fixed within the calling context. Furthermore, all in-line constraints must be specified within a single call to `randomize()`.

Policy classes are a technique for applying SystemVerilog constraints in a portable, reusable, and incremental manner, originally described by John Dickol [1] [2]. The operating mechanism leverages an aspect of "global constraints," the simultaneous solving of constraints across a set of random objects. Randomizing an object that contains policies also randomizes the policies. Meanwhile, the policies contain a reference back to the container. Consequently, the policy container is constrained by the policies it contains. Dickol's approach is illustrated by the following code.

```
1 class policy_base#(type ITEM=uvmm_object);
2     ITEM item;
3
4     virtual function void set_item(ITEM item);
5         this.item = item;
6     endfunction
7 endclass
8
9 class policy_list#(type ITEM=uvmm_object) extends policy_base#(ITEM);
10     rand policy_base#(ITEM) policy[$];
11
12     function void add(policy_base#(ITEM) pcy);
13         policy.push_back(pcy);
14     endfunction
15
16     function void set_item(ITEM item);
17         foreach(policy[i]) policy[i].set_item(item);
18     endfunction
19 endclass
```

Fig. 1: The `policy_base` and `policy_list` classes

These two base classes provide the core definitions for policies: **policy\_base** implements the hook back to the policy container, and **policy\_list** enables related policies to be organized into groups. Both classes are parameterized by a container object type, so a unique specialization will be required for each policy-enabled container. Policy containers like transactions, sequences, and configuration objects implement these classes to support flexible random steering. Below are examples of a generic transaction, **addr\_txn**, with a random address and size and some policy classes to constrain those attributes.

```

1  class addr_txn;
2      rand addr_t addr;
3      rand int    size;
4      rand policy_base#(addr_txn) policy[$];
5
6      constraint c_size {size inside {1, 2, 4};}
7
8      function void pre_randomize;
9          foreach(policy[i]) policy[i].set_item(this);
10         endfunction
11     endclass
12
13     class addr_policy_base extends policy_base#(addr_txn);
14         addr_range ranges[$];
15
16         function void add(addr_t min, addr_t max);
17             addr_range rng = new(min, max);
18             ranges.push_back(rng);
19         endfunction
20     endclass
21
22     class addr_permit_policy extends addr_policy_base;
23         rand int selection;
24
25         constraint c_addr_permit {
26             selection inside {[0:ranges.size()-1]};
27
28             foreach(ranges[i]) {
29                 if(selection == i) {
30                     item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
31                 }
32             }
33         }
34     endclass
35
36     class addr_prohibit_policy extends addr_policy_base;
37         constraint c_addr_prohibit {
38             foreach(ranges[i]) {
39                 !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
40             }
41         }
42     endclass

```

Fig. 2: The **addr\_txn** class and policies for constraining it

The **addr\_permit\_policy** and **addr\_prohibit\_policy** classes implement some policies for constraining **addr\_txn** addresses. Address ranges can be stored in the **ranges** array. The **addr\_permit\_policy** will choose one of the ranges at random and constrain the address to be within the range, while the **addr\_prohibit\_policy** will exclude addresses that fall

within any of the ranges in its list.

The final class shows how policies might be used. The `addr_constrained_txn` class extends `addr_txn` and defines two policies, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The `addr_constrained_txn` class then passes the local `pcy` list to the parent `policy` queue.

```
1 class addr_constrained_txn extends addr_txn;
2   function new;
3     addr_permit_policy    permit = new;
4     addr_prohibit_policy  prohibit = new;
5     policy_list#(addr_txn) pcy = new;
6
7     permit.add('h00000000', 'h0000FFFF');
8     permit.add('h10000000', 'h1FFFFFFF');
9     pcy.add(permit);
10
11    prohibit.add('h13000000', 'h130FFFFF');
12    pcy.add(prohibit);
13
14    this.policy = {pcy};
15  endfunction
16 endclass
```

Fig. 3: The `addr_constrained_txn` subclass

At this point, an instance of `addr_constrained_txn` can be created and randomized like normal, and the address will be constrained based on the embedded policies.

```
1 addr_constrained_txn txn = new();
2 txn.randomize();
```

Fig. 4: Randomizing an instance of `addr_constrained_txn`

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [3]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [4]. Additionally, McClish defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing of the upgraded policy package implementation will also be discussed and resolved.

## II. PROBLEM #1: PARAMETERIZED POLICIES

The first problem with the above policy implementation is that because `policy_base` is parameterized to the class it constrains, different specializations cannot be grouped and indexed. The awkward consequences of this limitation become apparent when using policies with a class hierarchy. If you extend a class and add a new random field then you need a new policy type to constrain that field. The new policy type requires its own policy list, and the new list must be traversed and mapped back to the container during `pre_randomize()`.

Imagine a complex class hierarchy with several layers of inheritance and extension, and constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [5]). Using policies becomes cumbersome in this case because each class layer requires a unique family of constraints organized into a distinct list. This stratification of polices places a burden on users to know which layer of the class hierarchy defines each attribute they want to constrain, and the name of the associated policy list for the matching policy type. For example, extending the `addr_txn` class to create a version with parity checking results in a class hierarchy that looks like this:

```

1 class addr_txn;
2   // ... unchanged from previous example
3 endclass
4
5 class addr_p_txn extends addr_txn;
6   rand bit parity;
7   rand bit parity_err;
8   rand policy_base#(addr_p_txn) addr_p_policy[$];
9
10  constraint c_parity_err {
11    soft (parity_err == 0);
12    (parity_err) ^ ($countones({addr, parity}) == 1);
13  }
14
15  function void pre_randomize;
16    super.pre_randomize();
17    foreach(addr_p_policy[i]) addr_p_policy[i].set_item(this);
18  endfunction
19 endclass
20
21 class addr_constrained_txn extends addr_p_txn;
22  function new;
23    addr_permit_policy      permit_p = new;
24    addr_prohibit_policy    prohibit_p = new;
25
26    // definition to follow in a later example - constrains parity_err bit
27    addr_parity_err_policy  parity_err_p = new;
28
29    policy_list#(addr_txn)  addr_pcy_lst = new;
30    policy_list#(addr_p_txn) addr_p_pcy_lst = new;
31
32    permit_p.add('h00000000, 'h0000FFFF);
33    permit_p.add('h10000000, 'h1FFFFFFF);
34    addr_pcy_lst.add(permit_p);
35
36    prohibit_p.add('h13000000, 'h130FFFFFF);
37    addr_pcy_lst.add(prohibit_p);
38
39    parity_err_p.set(1'b1);
40    addr_p_pcy_lst.add(parity_err_p);
41
42    this.addr_policy      = {addr_pcy_lst};
43    this.addr_p_policy    = {addr_p_pcy_lst};
44  endfunction
45 endclass

```

Fig. 5: Modified address transaction classes with a parity checking subclass and policies included

Scaling this implementation results in a lot of repeated boilerplate code and is not very intuitive to use. Each additional subclass in a hierarchy only increases the chaos and complexity of implementing and using policies effectively.

The solution to this problem is to replace the parameterized policy base with a non-parameterized base and a parameterized extension. We chose an interface class as our non-parameterized base for the flexibility it offers over a virtual base class—specifically, our policies are bound only to implement the interface functions and not to extend a specific class implementation.

```

1 interface class policy;
2     pure virtual function void set_item(uvm_object item);
3 endclass
4
5 virtual class policy_imp#(type ITEM=uvm_object) implements policy;
6     protected rand ITEM m_item;
7
8     virtual function void set_item(uvm_object item);
9         if (!$cast(m_item, item)) begin
10             `uvm_warning("policy::set_item()", "Item/policy type mismatch")
11             this.m_item = null;
12             this.m_item.rand_mode(0);
13         end
14     endfunction: set_item
15 endclass: policy_imp
16
17 typedef policy policy_queue[$];

```

Fig. 6: The `policy` interface class and `policy_imp` class

A non-parameterized base enables all policies targeting a particular class hierarchy to be stored within a single common `policy_queue`. A parameterized template, `policy_imp`, implements the base interface and core functionality required by all policies.

One consequence of eliminating the parameter from our base type is that the policy-enabled container object, `item`, and its assignment function, `set_item()`, are no longer strongly typed. Here we make a small concession, using `uvm_object` as our default policy-enabled type. This means that all classes that implement our policies must derive from `uvm_object`, and we need to use dynamic casting to ensure that policies and their containers are type-compatible. In the example above, `item` is set to `null` and randomization is disabled when the cast fails, preventing runtime problems in the event that incompatible policies are applied.

Not much changes when it comes to defining policies; the address policies now extend `policy_imp` instead of `policy_base`, and the underlying constraints are written as implications so that they will not apply when `item` is missing.

```

1 class addr_policy extends policy_imp#(addr_txn);
2     // ... unchanged from previous example, with updated class extension
3 endclass
4
5 class addr_parity_err_policy extends policy_imp#(addr_p_txn);
6     protected bit parity_err;
7
8     constraint c_fixed_value {m_item != null -> m_item.parity_err == parity_err;}
9
10    function new(bit parity_err);
11        this.parity_err = parity_err;
12    endfunction
13 endclass
14
15 class addr_permit_policy extends addr_policy;
16     // same as before
17 endclass
18
19 class addr_prohibit_policy extends addr_policy;

```

```

20 // same as before
21 endclass

```

Fig. 7: Address policies updated to use the new policy implementation

However, the address transaction classes are simplified considerably. Only a single policy queue is required in the class hierarchy, and the vast majority of the boilerplate code has been eliminated, including all of the specialized lists. The `addr_txn` class now extends `uvm_object` to provide compatibility with the policy interface.

```

1 class addr_txn extends uvm_object;
2     rand policy_queue policies;
3     // policy is replaced with the above. All other members, constraints,
4     // and pre_randomize are unchanged from the previous example
5 endclass
6
7 class addr_p_txn extends addr_txn;
8     rand bit parity;
9     rand bit parity_err;
10    constraint c_parity_err {/*....*/}
11    // The local addr_p_policy and pre_randomize are removed. Everything
12    // else is unchanged from the previous example
13 endclass
14
15 class addr_constrained_txn extends addr_p_txn;
16     function new;
17         addr_permit_policy    permit_p    = new();
18         addr_prohibit_policy   prohibit_p  = new();
19         addr_parity_err_policy parity_err_p;
20
21         // only a single policy queue is necessary now
22         permit_p.add('h00000000, 'h0000FFFF);
23         permit_p.add('h10000000, 'h1FFFFFFF);
24         this.policies.push_back(permit_p);
25
26         prohibit_p.add('h13000000, 'h130FFFFFF);
27         this.policies.push_back(prohibit_p);
28
29         parity_err_p = new(1'b1);
30         this.policies.push_back(parity_err_p);
31     endfunction
32 endclass

```

Fig. 8: Address transaction classes updated to use the new policy implementation

### III. PROBLEM #2: DEFINITION LOCATION

The second problem with policies is “where do I define my policy classes?” This is not a complicated problem to solve; most users will likely wish to place their policy classes in a file or files close to the class they are constraining. However, directly embedding policy definitions within the class they constrain offers a myriad of benefits. Not only does this convention eliminate all guesswork about where to define and discover policies, but embedded policies also gain access to all members of their container class, including protected properties and methods! This privileged access enables policies to constrain attributes of a class that are not otherwise exposed, improving encapsulation.

To further optimize the organization of potentially large families of policies, we establish a convention of defining all policy classes within an embedded wrapper class called `POLICIES`. Each layer of a class hierarchy that implements policies will have its own embedded `POLICIES` wrapper, and individual `POLICIES` wrappers extend other wrappers in a manner

parallel to their container classes. This parallel inheritance pattern is shown below, with `addr_p_txn::POLICIES` extending `addr_txn::POLICIES`.

```
1 class addr_txn extends uvm_object;
2   // class members, constraints, and pre_randomize unchanged from previous
3
4   class POLICIES;
5     class addr_policy extends policy_imp#(addr_txn);
6       // ... unchanged from previous standalone class example
7     endclass
8
9     class addr_permit_policy extends addr_policy;
10      // ... unchanged from previous standalone class example
11    endclass
12
13    class addr_prohibit_p_policy extends addr_policy;
14      // ... unchanged from previous standalone class example
15    endclass
16  endclass: POLICIES
17 endclass
18
19 class addr_p_txn extends addr_txn;
20   protected rand bit parity_err;
21   // other class members and constraints unchanged from previous example
22
23   class POLICIES extends addr_txn::POLICIES;
24     class addr_parity_err_policy extends policy_imp#(addr_p_txn);
25       // ... unchanged from previous example
26     endclass
27
28     static function addr_parity_err_policy PARITY_ERR(bit value);
29       PARITY_ERR = new(value);
30     endfunction
31   endclass: POLICIES
32 endclass
33
34 class addr_constrained_txn extends addr_p_txn;
35   function new;
36     addr_constrained_txn::POLICIES::addr_permit_policy permit_p = new();
37     addr_constrained_txn::POLICIES::addr_prohibit_policy prohibit_p = new();
38
39     // policy constraint value setup unchanged from previous example
40
41     this.policies.push_back(
42       addr_constrained_txn::POLICIES::PARITY_ERR(1'b1)
43     );
44   endfunction
45 endclass
```

Fig. 9: Address transaction classes with embedded policies

This example also shows how we can define static constructor functions within `POLICIES` wrappers. This practice further reduces the cost of using policies since we can instantiate and initialize them with a single call, as demonstrated with the call to `addr_constrained_txn::POLICIES::PARITY_ERR()`. Note that although the `PARITY_ERR` constructor is defined in `addr_p_txn::POLICIES`, it is accessible through `addr_constrained_txn::POLICIES` because of the wrapper class in-

heritance. The `POLICIES::` scoping layer even helps to make code more readable and easy to understand.

What's more, the `parity_err` property has now been defined as `protected`, preventing anything but our `PARITY_ERR` policy from manipulating that "knob." In fact, a more advanced use of policies might define all members of a target class as protected, restricting the setting of fields exclusively through policies and reading through accessor functions, thus encouraging maximum encapsulation/loose coupling, which reduces the cost of maintaining and enhancing code and prevents bugs from cascading into classes that use policy-enabled classes.

#### IV. PROBLEM #3: BOILERPLATE OVERLOAD

The third problem with using policies is that policies are relatively expensive to define since you need at a minimum: a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints (property equals X), range constraints (property between Y and Z), or set membership (keyword `inside`) constraints, macros can be used to drastically reduce the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded `POLICIES` class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for complex constraints, if necessary.

```

1 // Fixed-value policy class and constructor macro
2 `define fixed_policy(POLICY, FIELD, TYPE) \
3 `m_fixed_policy_class(POLICY, FIELD, TYPE) \
4 `m_fixed_policy_constructor(POLICY, TYPE)
5
6 `define m_fixed_policy_class(POLICY, FIELD, TYPE) \
7     class POLICY``_policy extends base_policy; \
8     constraint c_fixed_value { \
9         (m_item != null) -> (m_item.FIELD == l_val); \
10    } \
11 \
12     function new(TYPE value); \
13         this.l_val = value; \
14     endfunction \
15 endclass: POLICY``_policy
16
17 `define m_fixed_policy_constructor(POLICY, TYPE) \
18     static function POLICY``_policy POLICY(TYPE value); \
19         POLICY = new(value); \
20     endfunction: POLICY

```

Fig. 10: Macros for setting up the embedded `POLICIES` class and a fixed value policy

This example includes a ``fixed_policy` macro, which wraps two additional macros responsible for creating a policy class and a static constructor for the class. This ``fixed_policy` example policy class lets you constrain a property to a fixed value. A more complete macro definition can be found in Appendix B. The appendix includes ``start_policies`, ``start_extended_policies`, and ``end_policies` macros that are used to create the embedded `POLICIES` class within the constrained class instead of using hard-coded `class` and `endclass` statements. They set up class inheritance as needed and create a local typedef for the `policy_imp` parameterized type.

```

1 class addr_txn extends uvm_object;
2     // class members, constraints, and pre_randomize unchanged from previous
3     ↪ example
4
5     `start_policies(addr_txn)
6         `include "addr_policies.svh"

```



```

6     `end_policies
7   endclass
8
9   class addr_p_txn extends addr_txn;
10    // class members and constraints unchanged from previous example
11
12    `start_extended_policies(addr_p_txn, addr_txn)
13      `fixed_policy(PARITY_ERR, parity_err, bit)
14    `end_policies
15  endclass
16
17  class addr_constrained_txn extends addr_p_txn;
18    // ... unchanged from previous example
19  endclass

```

Fig. 11: Simplified address transaction classes using the policy macros

The base `addr_txn` class has complex policies with a relationship between the `addr` and `size` fields, so rather than creating a policy macro that will only be used once, they can either be left as-is within the embedded policies class, or moved to a separate file and included with ``include` as was done here to keep the transaction class simple. The child parity transaction class is able to use the ``fixed_policy` macro to constrain the `parity_err` field. The constraint block remains the same as the previous example.

#### V. PROBLEM #4: UNEXPECTED POLICY REUSE BEHAVIOR AND OPTIMIZING FOR LIGHTWEIGHT POLICIES

A fourth problem occurred during our initial deployment of policies. We observed occasional unexpected behavior when attempting to re-randomize objects with policies. We didn't thoroughly characterize the behavior, but sometimes policies seemed to "remember" previous randomizations and wouldn't reapply their constraints during subsequent `randomize` calls. Results would clearly violate even simple policies.

Our policy architecture prioritizes scalability; policy classes are lightweight with a minimal footprint. Rather than investing effort to diagnose and work around the problem with reusing policies, we adopted a "use once and discard" approach, leaning into their disposable nature. It costs little to apply fresh policy instances before re-randomizing a target object. Following this strategy completely eliminated policy misbehavior.

To facilitate a safer form of policy reuse we introduced a `copy` method that returns a fresh policy instance initialized to the same state as the policy that implements it. We also doubled down on our use of static constructors to generate initialized policies.

```

1   static function addr_parity_err_policy PARITY_ERR(bit value);
2     PARITY_ERR = new(value);
3   endfunction
4   ...
5   this.policies = '{addr_constrained_txn::POLICIES::PARITY_ERR(1'b1)};

```

Fig. 12: Example static constructor function from the `PARITY_ERR` policy class

Passing array literals populated by policy instances from static constructors proved to be an excellent way to pack a lot of intent into little code. It also neatly worked around the reliability issues of reused policies.

#### VI. MORE IMPROVEMENTS TO THE POLICY PACKAGE

The examples presented so far are functional, but are lacking many features that would be useful in a real-world implementation. The following examples will present additional improvements to the policy package that will make it more practical and efficient to use.

##### A. Expanding the `policy` interface class

The following `policy` interface class adds additional methods for managing a policy.

```

1 interface class policy;
2
3     pure virtual function string name();
4     pure virtual function string type_name();
5     pure virtual function string description();
6     pure virtual function bit item_is_compatible(uvm_object item);
7     pure virtual function void set_item(uvm_object item);
8     pure virtual function policy copy();
9
10 endclass: policy

```

Fig. 13: Expanded **policy** interface class

The **name**, **description**, and **copy** methods are implemented by the policy (or policy macro) directly and provide reporting information useful when printing messages about the policy to the log for the former two, or specific behavior for making a copy for the latter. The remaining three methods are implemented by **policy\_imp** and are shared by all policies.

#### B. Better type safety checking and reporting in **policy\_imp** methods

Some of the benefits of above methods can be seen by examining the new **set\_item** method used by **policy\_imp**.

```

1 virtual function void set_item(uvm_object item);
2     if (item == null) begin
3         `uvm_error("policy::set_item()", "NULL item passed")
4
5     end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
6         `uvm_info(
7             "policy::set_item()",
8             $sformatf(
9                 "policy <%s> applied to item <%s>: %s",
10                this.name(), item.get_name(), this.description()
11            ),
12            UVM_FULL
13        )
14        this.m_item.rand_mode( 1 );
15
16    end else begin
17        `uvm_warning(
18            "policy::set_item()",
19            $sformatf(
20                "Cannot apply policy '%0s' of type '%0s' to target object '%0s' of
↪ incompatible type '%0s'",
21                this.name(), ITEM::type_name(), item.get_name(),
↪ item.get_type_name()
22            )
23        )
24        this.m_item = null;
25        this.m_item.rand_mode( 0 );
26    end
27 endfunction: set_item

```

Fig. 14: **set\_item** method from **policy\_imp**

The `set_item` method makes use of all the reporting methods to provide detailed log messages when `set_item` succeeds or fails. Additionally, the `item_is_compatible` is used before the `\$cast` method is called and the `rand_mode` state is kept consistent with the result of the cast.

### C. Replacing `policy_list` with `policy_queue`

Eagle-eyed readers might have noticed the lack of presence of a `policy_list` class in any of the examples above after migrating to the improved policy interface. Rather, a single typedef is all that is necessary to manage policies in a class.

```
1 typedef policy policy_queue[$];
```

Fig. 15: The `policy_queue` typedef

The `policy_queue` type is capable of storing any policy that implements the `policy` interface. The default queue methods are sufficient for aggregating policies, and in practice we found that using policy queues as containers was more efficient than `policy_list` instances. For example, for functions expecting a `policy_queue` argument we can directly pass in array literals populated by calls to static constructor functions, allowing us to define, initialize, aggregate, and pass policies all in a single line of code!

### D. Standardize policy implementations with the `policy_container` interface and `policy_object` mixin

The `policy_container` interface class defines a set of functions for managing policies using `policy_queue` arguments. These functions provide a simple, standard way to implement policies across a verification environment.

```
1 interface class policy_container;
2
3     // Queries
4     pure virtual function bit has_policies();
5
6     // Assignments
7     pure virtual function void set_policies(policy_queue policies);
8     pure virtual function void add_policies(policy_queue policies);
9     pure virtual function void clear_policies();
10
11    // Access
12    pure virtual function policy_queue get_policies();
13
14    // Copy
15    pure virtual function policy_queue copy_policies();
16
17 endclass: policy_container
```

Fig. 16: The `policy_container` interface class

The `policy_object` mixin implements the `policy_container` interface and contains a protected `policy_queue` for managing policies (a complete example is available in Appendix A).

```
1 class policy_object #(type BASE=uvm_object) extends BASE implements
2   ↪ policy_container;
3
4     protected policy_queue m_policies;
5
6     // Queries
7     virtual function bit has_policies();
8         // returns true/false based on size of m_policies
```

```

8     endfunction: has_policies
9
10    // Assignments
11    virtual function void set_policies(policy_queue policies);
12        // sets m_policies to a new queue of policies
13    endfunction: set_policies
14
15    virtual function void add_policies(policy_queue policies);
16        // adds new policies to m_policies
17    endfunction: add_policies
18
19    virtual function void clear_policies();
20        // clears m_policies
21    endfunction: clear_policies
22
23    // Access
24    virtual function policy_queue get_policies();
25        // return a handle to m_policies
26    endfunction: get_policies
27
28    // Copy
29    virtual function policy_queue copy_policies();
30        // a copy of m_policies
31    endfunction: copy_policies
32 endclass: policy_object

```

Fig. 17: A `policy_object` base class implementation

The `policy_object` mixin can be applied to any class that might benefit from the use of policies, as seen in the following examples.

```

1 // Use policy_object for transactions
2 class base_txn extends policy_object #(uvm_sequence_item);
3
4 // Use policy_object for sequences
5 class base_seq #(type REQ=uvm_sequence_item, RSP=REQ) extends policy_object #(
6     ↪ uvm_sequence #(REQ, RSP) );
7
8 // Use policy_object for configuration objects
9 class cfg_object extends policy_object #(uvm_object);

```

Fig. 18: Example classes using the `policy_object` mixin

#### E. Protecting the policy queue enforces loosely coupled code

A subtle but significant additional benefit to using a base `policy_object` class along with the `policy_container` API is the ability to mark the container's policy queue as protected and prevent direct access to it.

The original implementation called `set_item` on each policy during the `pre_randomize` stage. That was necessary because the `policy_queue` was public, so there was nothing to prevent callers from adding policies without linking them to the target class.

Using an interface class and making the implementation private means the callers may only set or add policies using the available interface class routines, and because those routines are solely responsible for applying policies, they can also check compatibility (and filter incompatible policies) and call `set_item` immediately. This can be seen in the example `policy_object` implementation above, in the usage of the protected function `try_add_policy`.

By calling `set_item` when the policy is added to the queue, all policies will be associated with the target item automatically, so there is no need to do it during `pre_randomize`. This eliminates an easily-overlooked requirement for classes extending `policy_object` to make sure they call `super.pre_randomize()` in their local `pre_randomize` function.

## VII. CONCLUSION

The improvements to the policy package presented in this paper provide a more robust and efficient implementation of policy-based constraints for SystemVerilog. The policy package is now capable of managing constraints across an entire class hierarchy, and the policy definitions are tightly paired with the class they constrain. The use of macros reduces the expense of defining common policies, while still allowing great flexibility in any custom policies necessary.

The complete policy package is available in Appendix A, with additional macro examples available in Appendix B. A functional package is also available for download [6] which can be included directly in a project to start using policies immediately.

## REFERENCES

- [1] J. Dickol, "SystemVerilog Constraint Layering via Reusable Randomization Policy Classes," DVCon, 2015.
- [2] J. Dickol, "Complex Constraints: Unleashing the Power of the VCS Constraint Solver," SNUG Austin, September 29, 2016.
- [3] K. Vasconcellos, J. McNeal, "Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution," DVCon, 2021.
- [4] C. McClish, "Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins," DVCon, 2021.
- [5] J. Vance, J. Montesano, M. Litterick, J. Sprott, "Be a Sequence Pro to Avoid Bad Con Sequences," DVCon, 2019.
- [6] D. Mills, C. Haldane, "policy\_pkg Source Code," [https://github.com/DillanCMills/policy\\_pkg](https://github.com/DillanCMills/policy_pkg).

APPENDIX A  
POLICY PACKAGE

A. *policy\_pkg.sv*

```
1  `include "uvm_macros.svh"
2
3  package policy_pkg;
4
5      import uvm_pkg::*;
6
7      `include "policy.svh"
8
9      typedef policy policy_queue[$];
10
11     `include "policy_imp.svh"
12
13     `include "policy_container.svh"
14     `include "policy_object.svh"
15
16 endpackage: policy_pkg
```

B. *policy.svh*

```
1  interface class policy;
2
3      pure virtual function string name();
4      pure virtual function string type_name();
5      pure virtual function string description();
6      pure virtual function bit item_is_compatible(uvm_object item);
7      pure virtual function void set_item(uvm_object item);
8      pure virtual function policy copy();
9
10 endclass: policy
```

C. *policy\_imp.svh*

```
1  virtual class policy_imp #(type ITEM=uvm_object) implements policy;
2
3      protected rand ITEM m_item;
4
5      pure virtual function string name();
6      pure virtual function string description();
7      pure virtual function policy copy();
8
9      virtual function string type_name();
10         return (ITEM::type_name);
11     endfunction: type_name
12
13     virtual function bit item_is_compatible(uvm_object item);
14         ITEM local_item;
15
16         return ((item != null) && ($cast(local_item, item)));
```

```

17     endfunction: item_is_compatible
18
19     virtual function void set_item(uvm_object item);
20         if (item == null)
21             `uvm_error("policy::set_item()", "NULL item passed")
22
23         else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
24             `uvm_info(
25                 "policy::set_item()",
26                 $sformatf(
27                     "policy <%s> applied to item <%s>: %s",
28                     this.name(), item.get_name(), this.description()
29                 ),
30                 UVM_FULL
31             )
32             this.m_item.rand_mode( 1 );
33
34         end else begin
35             `uvm_warning(
36                 "policy::set_item()",
37                 $sformatf(
38                     "Cannot apply policy '%0s' of type '%0s' to target object '%0s'
↪ of incompatible type '%0s'",
39                     this.name(), this.type_name(), item.get_name(),
↪ item.get_type_name()
40                 )
41             )
42             this.m_item = null;
43             this.m_item.rand_mode( 0 );
44         end
45     endfunction: set_item
46
47 endclass: policy_imp

```

#### D. policy\_container.svh

```

1     interface class policy_container;
2
3         // Queries
4         pure virtual function bit has_policies();
5
6         // Assignments
7         pure virtual function void set_policies(policy_queue policies);
8         pure virtual function void add_policies(policy_queue policies);
9         pure virtual function void clear_policies();
10
11        // Access
12        pure virtual function policy_queue get_policies();
13
14        // Copy
15        pure virtual function policy_queue copy_policies();
16
17    endclass: policy_container

```

## E. policy\_object.svh

```
1 class policy_object #(type BASE=uvvm_object) extends BASE implements
  ↳ policy_container;
2
3   protected rand policy_queue m_policies;
4
5   // Queries
6   virtual function bit has_policies();
7       return (this.m_policies.size() > 0);
8   endfunction: has_policies
9
10  // Assignments
11  virtual function void set_policies(policy_queue policies);
12      if(this.has_policies())
13          `uvm_warning( "policy", "set_policies() replacing existing policies" )
14
15      this.m_policies = {};
16      foreach(policies[i]) try_add_policy(policies[i]);
17  endfunction: set_policies
18
19  virtual function void add_policies(policy_queue policies);
20      foreach(policies[i]) try_add_policy(policies[i]);
21  endfunction: add_policies
22
23  virtual function void clear_policies();
24      if (this.has_policies())
25          `uvm_info("policy", $sformatf("clearing [%0d] policies from %s",
  ↳ this.m_policies.size(), this.get_name()), UVM_FULL)
26
27      this.m_policies = {};
28  endfunction: clear_policies
29
30  // Access
31  virtual function policy_queue get_policies();
32      return (this.m_policies);
33  endfunction: get_policies
34
35  // Copy
36  virtual function policy_queue copy_policies();
37      copy_policies = {};
38      foreach(this.m_policies[i])
39          copy_policies.push_back(this.m_policies[i].copy());
40  endfunction: copy_policies
41
42  protected function void try_add_policy(policy new_policy);
43      if (new_policy.item_is_compatible(this)) begin
44          `uvm_info("policy", $sformatf("adding policy %s to %s",
  ↳ new_policy.name, this.get_name()), UVM_FULL)
45
46          new_policy.set_item(this);
47          this.m_policies.push_back(new_policy);
48      end else
```



```
49         `uvm_warning("policy", $sformatf("policy %s not compatible with target
↔ %s", new_policy.name, this.get_name()))
50     endfunction: try_add_policy
51
52     // Constructor
53     function new(string name="policy_object");
54         super.new(name);
55     endfunction: new
56
57 endclass: policy_object
```

## APPENDIX B POLICY MACROS

### A. *policy\_macros.svh*

```
1  `ifndef __POLICY_MACROS__
2  `define __POLICY_MACROS__
3
4  //=====
5  // Embedded POLICIES class macros
6  //=====
7
8  `define start_policies(cls) \
9  class POLICIES; \
10 `m_base_policy(cls)
11
12 `define start_extended_policies(cls, parent) \
13 class POLICIES extends parent::POLICIES; \
14 `m_base_policy(cls)
15
16 `define end_policies \
17 endclass: POLICIES
18
19 `define m_base_policy(cls) \
20     typedef policy_imp#cls) base_policy;
21
22
23 //=====
24 // Policy template macros
25 //=====
26
27 `include "constant_policy.svh"
28 `include "fixed_policy.svh"
29 `include "member_policy.svh"
30 `include "range_policy.svh"
31
32 `endif
```

### B. *constant\_policy.svh*

```
1  // Full policy definition
2  `define constant_policy(POLICY, FIELD, TYPE, CONST) \
3  `m_const_policy_class(POLICY, FIELD, TYPE, CONST) \
4  `m_const_policy_constructor(POLICY)
5
6  // Policy class definition
7  // Note - type casting within the constraint expression requires TYPE to be a
8  // single keyword (e.g. int, addr_t), not a vector expression (e.g. bit [31:0])
9  `define m_const_policy_class(POLICY, FIELD, TYPE, CONST) \
10     class POLICY`_policy extends base_policy; \
11     constraint c_policy_constraint { \
12         (m_item != null) -> (m_item.FIELD == TYPE'(CONST)); \
13     } \
14
```

```

15     function new();
16     endfunction: new
17
18     virtual function string name();
19         return (`"POLICY`");
20     endfunction: name
21
22     virtual function string description();
23         return (`"(FIELD==CONST)`");
24     endfunction: description
25
26     virtual function POLICY` `_policy copy();
27         copy = new();
28     endfunction: copy
29     endclass: POLICY` `_policy
30
31 // Policy constructor definition
32 `define m_const_policy_constructor(POLICY)
33     static function POLICY` `_policy POLICY();
34         POLICY = new();
35     endfunction: POLICY

```

### C. fixed\_policy.svh

```

1 // Full policy definition
2 `define fixed_policy(POLICY, FIELD, TYPE, RADIX="%0p")
3     `m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX)
4     `m_fixed_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
8     class POLICY` `_policy extends base_policy;
9         typedef TYPE    l_field_t;
10
11         local TYPE    l_val;
12         local string  l_radix=RADIX;
13
14         constraint c_policy_constraint {
15             (m_item != null) -> (m_item.FIELD == l_val);
16         }
17
18         function new(TYPE value, string radix=RADIX);
19             this.set_value(value);
20             this.set_radix(radix);
21         endfunction: new
22
23         virtual function string name();
24             return (`"POLICY`");
25         endfunction: name
26
27         virtual function string description();
28             return ({
29                 `(FIELD==`,
30                 $sformatf(l_radix, l_val),

```

```

31         ")"
32     });
33     endfunction: description
34
35     virtual function POLICY``_policy copy();
36         copy = new(l_val, l_radix);
37     endfunction: copy
38
39     virtual function void set_value(TYPE value);
40         this.l_val = value;
41     endfunction: set_value
42
43     virtual function TYPE get_value();
44         return (this.l_val);
45     endfunction: get_value
46
47     virtual function void set_radix(string radix);
48         this.l_radix = radix;
49     endfunction: set_radix
50
51     virtual function string get_radix();
52         return (this.l_radix);
53     endfunction: get_radix
54 endclass: POLICY``_policy
55
56 // Policy constructor definition
57 `define m_fixed_policy_constructor(POLICY, TYPE, RADIX="%0p")
58     static function POLICY``_policy POLICY(TYPE value, string radix=RADIX);
59         POLICY = new(value, radix);
60     endfunction: POLICY

```

#### D. member\_policy.svh

```

1 // Full policy definition
2 `define member_policy(POLICY, FIELD, TYPE, RADIX="%0p")
3     `m_member_policy_class(POLICY, FIELD, TYPE, RADIX)
4     `m_member_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_member_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
8     class POLICY``_policy extends base_policy;
9         typedef TYPE          l_field_array_t[];
10
11         local l_field_array_t  m_values;
12         local bit              l_exclude;
13         local string           m_radix=RADIX;
14
15         constraint c_policy_constraint {
16             (m_item != null) ->
17                 ((l_exclude) ^ (m_item.FIELD inside {m_values}));
18         }
19
20         function new(
21             l_field_array_t values,

```

```

22         bit          exclude=1'b0,
23         string       radix=RADIX
24     );
25     this.set_values(values);
26     this.set_exclude(exclude);
27     this.set_radix(radix);
28     endfunction: new
29
30     virtual function string name();
31         return (`"POLICY`");
32     endfunction: name
33
34     virtual function string description();
35         string values_str = "";
36
37         foreach(m_values[i])
38             values_str = {
39                 values_str,
40                 $sformatf(m_radix, m_values[i]),
41                 i == m_values.size()-1 ? "" : ", ";
42
43         return ({
44             `(FIELD `",
45             l_exclude ? "outside {" : "inside {" ,
46             values_str,
47             "}")"
48         });
49     endfunction: description
50
51     virtual function POLICY` `_policy copy();
52         copy = new(m_values, l_exclude, m_radix);
53     endfunction: copy
54
55     virtual function void set_values(l_field_array_t values);
56         this.m_values = values;
57     endfunction: set_values
58
59     virtual function l_field_array_t get_values();
60         l_field_array_t l_array;
61         foreach (this.m_values[i])
62             l_array[i] = this.m_values[i];
63         return (l_array);
64     endfunction: get_values
65
66     virtual function void set_exclude(bit exclude);
67         this.l_exclude = exclude;
68     endfunction: set_exclude
69
70     virtual function bit get_exclude();
71         return (this.l_exclude);
72     endfunction: get_exclude
73
74     virtual function void set_radix(string radix);
75         this.m_radix = radix;

```

```

76     endfunction: set_radix                                     \
77                                                         \
78     virtual function string get_radix();                   \
79         return (this.m_radix);                             \
80     endfunction: get_radix                                   \
81 endclass: POLICY` `_policy
82
83 // Policy constructor definition
84 `define m_member_policy_constructor(POLICY, TYPE, RADIX="%0p") \
85     typedef TYPE POLICY` `_array_t[];                     \
86     static function POLICY` `_policy POLICY(              \
87         POLICY` `_array_t values,                         \
88         bit          exclude=1'b0,                       \
89         string       radix=RADIX                          \
90     );                                                     \
91     POLICY = new(values, exclude, radix);                 \
92 endfunction: POLICY

```

### E. range\_policy.svh

```

1 // Full policy definition
2 `define range_policy(POLICY, FIELD, TYPE, RADIX="%0p") \
3     `m_range_policy_class(POLICY, FIELD, TYPE, RADIX) \
4     `m_range_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_range_policy_class(POLICY, FIELD, TYPE, RADIX="%0p") \
8     class POLICY` `_policy extends base_policy; \
9         local TYPE          l_low; \
10        local TYPE          l_high; \
11        local bit          l_exclude; \
12        local string       l_radix=RADIX; \
13
14        constraint c_policy_constraint { \
15            (m_item != null) -> ( \
16                (l_exclude) ^ \
17                (m_item.FIELD >= l_low && m_item.FIELD <= l_high) \
18            ); \
19        } \
20
21        function new( \
22            TYPE low, \
23            TYPE high=low, \
24            bit exclude=1'b0, \
25            string radix=RADIX \
26        ); \
27            this.set_range(low, high); \
28            this.set_exclude(exclude); \
29            this.set_radix(radix); \
30        endfunction: new \
31
32        virtual function string name(); \
33            return (~"POLICY` `"); \
34        endfunction: name

```

```

35 virtual function string description();
36     return ({
37         ~(FIELD `",
38         l_exclude ? "outside [" : "inside [",
39         $sformatf(l_radix, l_low),
40         ", ",
41         $sformatf(l_radix, l_high),
42         "]"
43     });
44 endfunction: description
45
46 virtual function POLICY``_policy copy();
47     copy = new(l_low, l_high, l_exclude, l_radix);
48 endfunction: copy
49
50 virtual function void set_range(TYPE low, TYPE high);
51     if (low <= high) begin
52         this.l_low = low;
53         this.l_high = high;
54     end else begin
55         this.l_low = high;
56         this.l_high = low;
57     end
58 endfunction: set_range
59
60 virtual function TYPE get_low();
61     return (this.l_low);
62 endfunction: get_low
63
64 virtual function TYPE get_high();
65     return (this.l_high);
66 endfunction: get_high
67
68 virtual function void set_exclude(bit exclude);
69     this.l_exclude = exclude;
70 endfunction: set_exclude
71
72 virtual function bit get_exclude();
73     return (this.l_exclude);
74 endfunction: get_exclude
75
76 virtual function void set_radix(string radix);
77     this.l_radix = radix;
78 endfunction: set_radix
79
80 virtual function string get_radix();
81     return (this.l_radix);
82 endfunction: get_radix
83 endclass: POLICY``_policy
84
85 // Policy constructor definition
86 `define m_range_policy_constructor(POLICY, TYPE, RADIX="%0p")
87     static function POLICY``_policy POLICY(
88

```

```
89     TYPE low, \
90     TYPE high=low, \
91     bit exclude=1'b0, \
92     string radix=RADIX \
93 ); \
94     POLICY = new(low, high, exclude, radix); \
95 endfunction: POLICY
```